task TOGGLE;
  input integer TIMES;
  repeat (TIMES) begin
    #(PERIOD / 2) CLK = 1'b0;
    #(PERIOD / 2) CLK = 1'b1;
  end
endtask
FIGURE 6.1 A task used to create a clock signal


module taskdemo;
  parameter WIDTH = 4;
  parameter PERIOD = 10.0;
  reg [WIDTH - 1 : 0] COUNT, DATA;
  reg CLK, RST, LOAD;

  always @(posedge CLK or negedge RST) begin
    if (!RST) COUNT <= 'b0;
    else begin
      if (LOAD) COUNT <= DATA;
      else COUNT <= COUNT + 1;
    end
  end

  initial TOGGLE(12); //Parallel execution
  
  initial begin
    //TOGGLE(12); //Serial execution
    DATA = 4'ha; 
    RST = 1'b1; LOAD = 1'b0;
    #PERIOD RST = 1'b0;
    #(2*PERIOD) RST = 1'b1;
    #(2*PERIOD) LOAD = 1'b1;
    #PERIOD LOAD = 1'b0;
  end

  task TOGGLE; //Operates on local variable CLK
    input integer TIMES;
    repeat (TIMES) begin
      #(PERIOD / 2) CLK = 1'b0;
      #(PERIOD / 2) CLK = 1'b1;
    end
  endtask
endmodule
FIGURE 6.2 Calling task TOGGLE in series and in parallel



task TOGGLE;
  input integer TIMES;
  output reg CLK;
  repeat (TIMES) begin
    #(PERIOD / 2) CLK = 1'b0;
    #(PERIOD / 2) CLK = 1'b1;
  end
endtask
FIGURE 6.5 Task with a local variable





module taskdemo;
  parameter WIDTH = 4;
  parameter PERIOD = 10.0;
  reg [WIDTH - 1 : 0] COUNT, DATA;
  reg CLK, RST, LOAD;

  always @(posedge CLK or negedge RST) begin
    if (!RST) COUNT <= 'b0;
    else begin
      if (LOAD) COUNT <= DATA;
      else COUNT <= COUNT + 1;
    end
  end

//Task now has both an input and an output
  initial TOGGLE(12, CLK); //Parallel execution
  
  initial begin
    //TOGGLE(12); //Serial execution
    DATA = 4'ha; 
    RST = 1'b1; LOAD = 1'b0;
    #PERIOD RST = 1'b0;
    #(2*PERIOD) RST = 1'b1;
    #(2*PERIOD) LOAD = 1'b1;
    #PERIOD LOAD = 1'b0;
  end

  task TOGGLE; //Operates on local variable CLK
    input integer TIMES;
    output reg CLK;
    repeat (TIMES) begin
      #(PERIOD / 2) CLK = 1'b0;
      #(PERIOD / 2) CLK = 1'b1;
    end
  endtask
  
endmodule
FIGURE 6.6 Call to task TOGGLE with a local variable




logic POSITIVE, NEGATIVE;
logic [15:0] DATABUS;

  task automatic PARITY([15:0] DBUS, output POSP, NEGP);
    @(posedge CLK)
      #1 POSP = ^DBUS; //output positive parity
      #3 NEGP = ~^DBUS; //output negative parity
  endtask
  
  always_ff @(posedge CLK)
    PARITY(DATABUS, POSITIVE, NEGATIVE);
FIGURE 6.9 Delays in a task


logic POSITIVE, NEGATIVE;
logic [15:0] DATABUS;

  task automatic PARITY([15:0] DBUS, output POSP, NEGP);
      #1 POSP = ^DBUS; //output positive parity
      #3 NEGP = ~^DBUS; //output negative parity
  endtask
  
  always_ff @(posedge CLK)
    PARITY(DATABUS, POSITIVE, NEGATIVE);
FIGURE 6.10 Synthesizable sequential circuit with a task


  task handshake();
    wait (READY);
    @(posedge CLK)
      VALID = 1'b1;
      DATA = RAWDATA;
      ADDRESS = RAWADDR;
    wait (ACK);
    @(posedge CLK) VALID = 1'b0;
  endtask
FIGURE 6.11 Using a task to execute repetitive series of commands




task look_for_X([31:0] DATABUS, [23:0] ADDRBUS);
  if (DATABUS === 'bz) begin
    $display("Data Bus is High Impedance");
    return;
  end
  else if (DATABUS == 'bx) begin
    $display ("Data Bus is unknown");
    return;
  end
  else begin
    case (ADDRBUS)
      0: if (DATABUS != 32'h0) FLAG = 1'b1;
      //rest of test cases go here
    endcase
  end
endtask
FIGURE 6.12 Task with return statements



module pargen (input [15:0] DATA, output logic POSP, NEGP);
  
  function automatic [1:0] calcparity;
    input [15:0] BUS;
    logic P, N;
    P = ^BUS;
    N = ~^BUS;
    calcparity = {P, N};
  endfunction
  
  always_comb {POSP, NEGP} = calcparity(DATA);  
endmodule
FIGURE 6.13 A function returning two one-bit values concatenated together



function byte scramble;
  input byte DIN;
  for (int I = 0; I < 8; I++)
    if (I % 2)
      scramble[I] = DIN[I/2];
    else
      scramble[I] = DIN[I/2 + 4];
endfunction
FIGURE 6.14 A function returning an eight-bit value




function automatic void calcp2(input [15:0] BUS, output logic P, N);
    P = ^BUS;
    N = ~^BUS;
  endfunction
  
  logic POS, NEG;
  logic [15:0] DATA;
  always_comb calcp2(DATA, POS, NEG);
FIGURE 6.15 A function with outputs and a call to that function


function void JUSTDISPLAY;
  $display("ERROR: Unexpected Output Value");
endfunction

module JDF;
  reg A, B;
  initial begin
    A = 0; B = 0;
    #1 B = 1;
    #1 A = 1; B = 0;
    #1 B = 1;
    #1;
  end
  
  always @(A, B)
    if (A != B) JUSTDISPLAY;
endmodule
FIGURE 6.16 A void function having no arguments or return value



function void ERRORCHECK(input integer A, B);
  if (A != B) $display(“ERROR: at time %d A = %b, B = %b”, $time, A, B);
endfunction
FIGURE 6.17 A void function with arguments but no outputs or return value




module returndemo(input byte DATA_IN, output longint ANSWER);
  
  function automatic longint factorial(input byte DATA_IN);
    if (DATA_IN < 2) return 1;
    else if (DATA_IN == 2) return 2;
    else factorial = DATA_IN * (factorial(DATA_IN - 1));
    endfunction
  
  always_comb ANSWER = factorial(DATA_IN);
endmodule
FIGURE 6.18 A function with return statements


module paramfunc(DATA, PARITY);
  parameter WIDTH = 32;
  input [WIDTH - 1 : 0] DATA;
  output reg PARITY;

  //#8 is a delay, not a parameter redefinition
  always @(DATA) PARITY = #(8) calc_parity(DATA);

function calc_parity; //Uses module parameter
  input reg[WIDTH - 1 : 0] DATA; //Width is 32 here too
  calc_parity = ^DATA;
endfunction

endmodule
FIGURE 6.19 Function using a parameter from the calling module


module paramfunc2(DATA, PARITY);
  parameter WIDTH = 32;
  input [WIDTH - 1 : 0] DATA;
  output reg PARITY;

always @(DATA) PARITY = #(32) calc_parity(DATA);

//Local declaration of WIDTH will not be over-ridden
function calc_parity;
  parameter WIDTH = 8;
  input reg[WIDTH - 1 : 0] DATA; //Now WIDTH is 8.
  calc_parity = ^DATA; //Parity calculated only over 8 bits
endfunction

endmodule
FIGURE 6.20 Defective use of a parameter in a function

package parpackage;
  //depends on module definitions: will not compile
  function automatic void parity (input [SIZE – 1 : 0] DATA,
    output logic P, N);
      P = ^DATA;
      N = ~^DATA;
    endfunction
endpackage
Figure 6.21 Package with a parameter. Will not compile



/*This does not work as written. It can easily be modified to
use include, which will then work.*/
module include_or_import #(SIZE = 8) (input [SIZE - 1 : 0] DATA,
  output logic POS, NEG);
  import parpackage::*;
  always_comb parity(DATA, POS, NEG);
endmodule
Figure 6.22 Module will fail to compile when it fails to find the package


module include_or_import #(SIZE = 8) (input [SIZE - 1 : 0] DATA,
  output logic POS, NEG);
  //include, don't import, when subroutine has dependencies
  include parpackage.sv;
  always_comb parity(.D(DATA), .P(POS), .N(NEG));
endmodule
FIGURE 6.23 Including a subroutine from a different file




module import_subs(output logic P, N);
  parameter SIZE = 8;
  parameter PERIOD = 10;
  //convert parameter PERIOD to variable PER so it can be referenced
  int PER = PERIOD;
  reg CLK;
  logic [SIZE - 1 : 0] DATA;
  import subs::*;
  initial TOGGLE(CLK, PER, 12);
  always_ff @(posedge CLK) par(DATA, P, N);
  initial begin
    DATA = 0;
    #PERIOD DATA = 1;
    #PERIOD DATA = 2;
    #PERIOD DATA = 3;
    #PERIOD DATA = 4;
    #PERIOD DATA = 5;
  end
endmodule

package subs;
task automatic TOGGLE(ref CLK, ref int PER, input integer TIMES);
    repeat (TIMES) begin
      #(PER / 2) CLK = 1'b0;
      #(PER / 2) CLK = 1'b1;
    end
endtask

/*SIZE parameter cannot be referenced: a value is hard coded
in the function, limiting its utility. A variable cannot be
used to set the size of another variable, so the workaround
used in task TOGGLE will not work here.*/
function automatic void par(ref [7:0] DATA, output logic P, N);
    P = ^DATA;
    N = ~^DATA;
endfunction
endpackage
FIGURE 6.24 Using references in packaged subroutines






interface BUS; //interface and endinterface are SV keywords
	logic [7:0] ADDR;
	wire [7:0] DATA; //bidirectional signal must be a net type
	logic OE, WS;
endinterface

module REGFILE (BUS ADBUS); //ADBUS is an instance of interface BUS
	reg [7:0] MEM [0:255];
  reg [7:0] DATA_REG;
  
  assign ADBUS.DATA = ADBUS.OE ? DATA_REG : 8'bz;

  always_comb
  DATA_REG = MEM[ADBUS.ADDR];

  always_ff @(posedge ADBUS.WS) MEM[ADBUS.ADDR] <= ADBUS.DATA;
endmodule
FIGURE 6.25 Register file with an interface


interface PBUS #(parameter WIDTH = 8, DEPTH = 8); 
	logic [DEPTH - 1:0] ADDR;
	wire [WIDTH - 1:0] DATA;
	logic OE, WS;
endinterface

module PREGFILE(PBUS ADBUS);
  parameter WIDTH = 16;
  parameter DEPTH = 16;
	reg [WIDTH - 1 : 0] MEM [0:2**DEPTH - 1];
  reg [WIDTH - 1 : 0] DATA_REG;
  
  assign ADBUS.DATA = ADBUS.OE ? DATA_REG : 'bz;

  always_comb DATA_REG = MEM[ADBUS.ADDR];

  always_ff @(posedge ADBUS.WS) MEM[ADBUS.ADDR] <= ADBUS.DATA;
endmodule

module TOP(input [7:0] ADDR, inout [5:0] DATA, input OE, WS);
  parameter WD = 4, DP = 6;
  PBUS #(.WIDTH(WD), .DEPTH(DP)) ADBUS(.*);
  PREGFILE #(.WIDTH(WD), .DEPTH(DP)) MEM(ADBUS);
endmodule
FIGURE 6.26 Scalable register file and interface in a top-level design module




interface PBUS_modport #(parameter WIDTH = 8, DEPTH = 8); 
	logic [DEPTH - 1:0] ADDR;
	wire [WIDTH - 1:0] DATA;
	logic OE, WS;
	modport MEMSIDE(input ADDR, OE, WS, inout DATA);
	modport PROCSIDE(output ADDR, OE, WS, inout DATA);
endinterface

module PREGFILE_modport(PBUS_modport.MEMSIDE ADBUS);
  parameter WIDTH = 16;
  parameter DEPTH = 16;
	reg [WIDTH - 1 : 0] MEM [0:2**DEPTH - 1];
  reg [WIDTH - 1 : 0] DATA_REG;
  
  assign ADBUS.DATA = ADBUS.OE ? DATA_REG : 'bz;

  always_comb DATA_REG = MEM[ADBUS.ADDR];

  always_ff @(posedge ADBUS.WS) MEM[ADBUS.ADDR] <= ADBUS.DATA;
endmodule

module TOP(input [7:0] ADDR, inout [5:0] DATA, input OE, WS);
  parameter WD = 4, DP = 6;
  PBUS_modort #(.WIDTH(WD), .DEPTH(DP)) ADBUS(.*);
  PREGFILE_modport #(.WIDTH(WD), .DEPTH(DP)) MEM(ADBUS);
endmodule
FIGURE 6.28 Register file with an interface using modports


interface CLOCKEDBUS(input CLK, RST);
  wire [7:0] DATA;
  logic [3:0] STATE;
  logic ENABLE, READ, WRITE;
  modport MASTER(input CLK, RST, output STATE, ENABLE, READ, 
    WRITE, inout DATA);
  modport SLAVE(input CLK, RST, STATE, ENABLE, READ, WRITE,
    inout DATA);
endinterface : CLOCKEDBUS
FIGURE 6.29 Interface with primary inputs




[bookmark: _GoBack]interface HANDSHAKE_INT(input CLK, RST, output logic VALID, READY);
  wire [7:0] DATA;
  modport SLAVE(input CLK, RST, READY, output VALID, inout DATA);
  modport MASTER(input CLK, RST, VALID, output READY, inout DATA);
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) VALID <= 1'b0;
    else
      if (READY) VALID <= 1'b1;
      else VALID <= 1'b0;
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) READY <= 1'b1;
    else
      if (VALID) READY <= 1'b0;
      else if (READY) READY <= 1'b1;
      else READY <= READY;
endinterface : HANDSHAKE_INT

module DRIVER(HANDSHAKE_INT.MASTER  I);
endmodule : DRIVER
module RECEIVER(HANDSHAKE_INT.SLAVE J);
endmodule : RECEIVER
  
module TOP(input CLK, RST, output VALID, READY);
	HANDSHAKE_INT	H(CLK, RST, VALID, READY);
	DRIVER	D(H.MASTER);
	RECEIVER R(H.SLAVE);
endmodule : TOP
FIGURE 6.30 Interface with behavioral code implementing handshake protocol
