module scalable_counter #(parameter SIZE = 4)
(input CLK, RST, output logic [SIZE - 1 : 0] COUNT);

always_ff @(posedge CLK, negedge RST)
  if (!RST) COUNT <= 0;
  else COUNT <= COUNT + 1;
endmodule

module scalable_counter ( CLK, RST, COUNT );
  output [3:0] COUNT;
  input CLK, RST;
  wire   N2, N3, N4, n1, n2, n5, n6, n7, n8;

  DFFARX1 \COUNT_reg[0]  ( .D(n2), .CLK(CLK), .RSTB(RST), .Q(COUNT[0]), .QN(n2) );
  DFFARX1 \COUNT_reg[1]  ( .D(N2), .CLK(CLK), .RSTB(RST), .Q(COUNT[1]), .QN(n8) );
  DFFARX1 \COUNT_reg[2]  ( .D(N3), .CLK(CLK), .RSTB(RST), .Q(COUNT[2]), .QN(n1) );
  DFFARX1 \COUNT_reg[3]  ( .D(N4), .CLK(CLK), .RSTB(RST), .Q(COUNT[3]), .QN(n7) );
  XOR2X1 U8 ( .IN1(n5), .IN2(n7), .Q(N4) );
  NAND2X0 U9 ( .IN1(n6), .IN2(COUNT[2]), .QN(n5) );
  XNOR2X1 U10 ( .IN1(n1), .IN2(n6), .Q(N3) );
  NOR2X0 U11 ( .IN1(n8), .IN2(n2), .QN(n6) );
  XOR2X1 U12 ( .IN1(n8), .IN2(n2), .Q(N2) );
endmodule
FIGURE 11.1 SystemVerilog behavioral code and synthesized netlist


`timescale 1 ps/1ps
/*Simple model of a 2-input AND gate
with constant 1 picosecond delay*/
module AND21A(A, B, Q);
  input A, B;
  output Q;
  and #1(Q, A, B);
endmodule

/*2-input AND gate model with rise and
fall time parameters*/
module AND21B(A, B, Q);
  input A, B;
  output Q;
  //1 picosecond rise time, 2 picosecond fall time
  and #(1, 2)(Q, A, B);
endmodule
FIGURE 11.2 Simple cell models with delays


`timescale 1ps/1ps
module TBUF1(IN, EN, Z);
  input IN, EN;
  output  Z;

  /*Rise, fall and turn off times*/
  bufif1 #(1, 2, 3) (Z,IN,EN);
endmodule
FIGURE 11.3 Tri-state buffer model with rise, fall, and turn-off time parameters


`timescale 1ps/1ps
/*Lumped delay AND-OR-INVERT cell*/
module AOI1(A, B, C, D, Q);
  input A, B, C, D;
  output Q;
  
  and (N1, A, B);
  and (N2, C, D);
  or  (N3, N1, N2);
  not #10 (Q, N3);
endmodule

/*Distributed delay AND-OR-INVERT cell*/
module AOI2(A, B, C, D, Q);
  input A, B, C, D;
  output Q;
  
  and #5 (N1, A, B);
  and #5 (N2, C, D);
  or  #4 (N3, N1, N2);
  not #1 (Q, N3);
endmodule
■FIGURE 11.4 AND-OR-INVERT cell with distributed and lumped delays


`timescale 1ps/1ps
module TBUF2a(IN, EN, Z);
  input IN, EN;
  output  Z;

  /*Rise, fall and turn-off times for
  best, typical and worst cases*/
  bufif1 #(1.6:3.2:4.4, 1.2:2.4:3.6, 1:2:3) (Z,IN,EN);
endmodule
FIGURE 11.5 Tri-state cell model with best case, typical case, and worst case operating conditions


`timescale 1ps/1ps
module TBUF2b(IN, EN, Z);
  input IN, EN;
  output  Z;
  bufif1 (Z,IN,EN);
  specify
  /*Rise, fall and turn-off times for best, typical 
  and worst cases*/
    (IN => Z) = (1.6:3.2:4.4, 1.2:2.4:3.6, 1:2:3);
  endspecify
endmodule
FIGURE 11.6 Tri-state cell with delay data in a specify block


/*AND-OR-INVERT cell with specify block*/
module AOI3(A, B, C, D, Q);
  input A, B, C, D;
  output Q;
  
  and (N1, A, B);
  and (N2, C, D);
  or (N3, N1, N2);
  not (Q, N3);
  specify
    (A, B *> Q) = 12, 10;
    (C, D *> Q) = 13, 11;
  endspecify
endmodule
FIGURE 11.7 AND-OR-INVERT cell with a full connection specify block




`timescale 1 ns/1ps
module AND2X1 (IN1,IN2,Q);

output  Q;
input   IN1,IN2;

and #1 (Q,IN2,IN1);

`ifdef functional
`else
specify
 specparam in1_lh_q_lh=52,in1_hl_q_hl=50,in2_lh_q_lh=59,in2_hl_q_hl=56;
 (        IN1 +=> Q) = (in1_lh_q_lh,in1_hl_q_hl);
 (        IN2 +=> Q) = (in2_lh_q_lh,in2_hl_q_hl);
    
endspecify
`endif

endmodule
FIGURE 11.8 AND cell with a specify block from Synopsys 90 nm library



module XNOR2X1 (IN1,IN2,Q);

output  Q;
input   IN1,IN2;

xnor #1 (Q,IN2,IN1);

`ifdef functional
`else
specify
 specparam in1_lh_q_hl=97,in1_hl_q_lh=114,in1_lh_q_lh=63,in1_hl_q_hl=64,
 in2_lh_q_hl=105,in2_lh_q_lh=78,in2_hl_q_lh=119,in2_hl_q_hl=80;
 if ((IN2==1'b0))
 (        IN1 -=> Q) = (in1_hl_q_lh,in1_lh_q_hl);
 if ((IN2==1'b1))
 (        IN1 +=> Q) = (in1_lh_q_lh,in1_hl_q_hl);
 if ((IN1==1'b0))
 (        IN2 -=> Q) = (in2_hl_q_lh,in2_lh_q_hl);
 if ((IN1==1'b1))
 (        IN2 +=> Q) = (in2_lh_q_lh,in2_hl_q_hl);
endspecify
`endif
endmodule
FIGURE 11.9 Conditional paths in a specify block from Synopsys 90 nm library


module FULL_ADDER(A, B, CI, S, CO);
  input A, B, CI;
  output wire S, CO;
  ADDR_CARRY (CO, A, B, CI);
  ADDR_SUM (S, A, B, CI);
  specify
    (A, B, CI *> CO) = 3, 2.4;
    (A, B, CI *> S) = 2.5, 2;
  endspecify
endmodule

// FULL ADDER CARRY-OUT TERM
primitive ADDR_CARRY (CO, A, B, CI);
output CO;
input A, B, CI;
table
// A    B   CI : CO
   0     0    ?  :  0;
   0     ?    0  :  0;
   ?     0    0  :  0;
   ?     1    1  :  1;
   1     ?    1  :  1;
   1     1    ?  :  1;
endtable
endprimitive

// FULL ADDER SUM TERM
primitive ADDR_SUM (S, A, B, CI);
output S;
input A, B, CI;
table
// A    B   CI  : S
   0     0    0  : 0;
   0     0    1  : 1;
   0     1    0  : 1;
   0     1    1  : 0;
   1     0    0  : 1;
   1     0    1  : 0;
   1     1    0  : 0;
   1     1    1  : 1;
endtable
endprimitive
FIGURE 11.10 Full adder model with user-defined primitives


module HiZ_Cell(A, B, EN, Z);
  input A, B, EN;
  output Z;
  
  //Q is output of UDP, not shown
  bufif1 (Z, Q, EN);
  Z_logic (Q, A, B);
  
  specify
    (A, B *> Z) = 2, 1.15, 1.7;
  endspecify
endmodule
FIGURE 11.11 Cell model with high-impedance mode


module COMPLEMENTARY(A, B, Q, Qb);
  input A, B;
  output Q, Qb;
  
  //Z is output of UDP, not shown
  buf (Q, Z);
  not (Qb, Z);
  C_logic (Z, A, B);
  
  specify
    (A, B +=> Q) = 2, 1;
    (A, B -=>Qb) = 1.9, 0.9;
  endspecify
endmodule
FIGURE 11.12 Cell model with complementary outputs



11.13 D latch cell model

FIGURE 11.14 D flipflop cell model

module D_latch(D, ENA, Q);
  input ENA, D;
  output Q;
  latch (Q, D, ENA);
  specify
    (D +=> Q) = (1.5, 1.2);
    (ENA +=> Q) = (1.7, 1.4);
  endspecify
endmodule
  
primitive latch_logic (Q, D, ENA);
output Q;	
reg Q; 	
input ENA, D;
table	
  //  D    ENA  Present  Next
  //                  state       state
      1    	1       : ? :         1 ;
      0    	1       : ? :         0 ;
      ?    	0       : ? :         - ;
endtable
endprimitive
FIGURE 11.13 D latch cell model




module DFF(CLOCK, D, Q);
  input CLOCK, D;
  output Q;
  DFF_logic(Q, CLOCK, D);
endmodule

primitive DFF_logic(Q, CLOCK, DATA);
  output Q;
  input CLOCK, DATA;
  reg Q; 	
  table	
//  CLOCK   DATA    Present   Next
//                  State     State
    (01)     1    :   ?   :    1  ;
    (01)     0    :   ?   :    0  ;
//Reduce pessimism for potential clocks
      p      1    :   1   :    1  ; 
      p      0    :   0   :    0  ;
//No output change on clock falling edges
    (?0)     ?    :   ?   :    -  ;
//No output change with steady-state clock
      ?    (??)   :   ?   :    -   ;
endtable
endprimitive
FIGURE 11.14 D flipflop cell model




module DFF(CLOCK, D, Q);
  input CLOCK, D;
  output Q;
  reg NOTIFIER;
  DFF_logic(Q, CLOCK, D, NOTIFIER);
  specify
    $setup(D, posedge CLOCK, 1, NOTIFIER);
    $hold(posedge CLOCK, D, 0.5, NOTIFIER);
  endspecify
endmodule

primitive DFF_logic(Q, CLOCK, DATA, NOTIFIER);
  output Q;
  input CLOCK, DATA, NOTIFIER;
  reg Q; 	
  table	
//  CLOCK   DATA  Notifier    Present  Next
//                            State    State
    (01)     1       ?      :   ?   :   1  ;
    (01)     0       ?      :   ?   :   0  ;
//Reduce pessimism for potential clock edges
     p       1       ?      :   1   :   1  ; 
     p       0       ?      :   0   :   0  ;
//No output change on clock falling edges
    (?0)     ?       ?      :   ?   :   -  ;
//No output change with steady-state clock
      ?    (??)      ?      :   ?   :   -  ;
//Set output unknown on any timing violation
      ?      ?       *      :   ?   :   x  ;
endtable
endprimitive
FIGURE 11.15 D flipflop cell model with notifier to detect timing violations




`timescale 1 ps / 1 ps
module JKFF(CLOCK, PRESET, RESET, J, K, Q);
  input CLOCK, PRESET, RESET, J, K;
  output Q;
  reg NOTIFIER;
  JKFF_logic(Q, CLOCK, J, K, PRESET, RESET, NOTIFIER);
  specify
    specparam tJSU = 12, tJHD = 6, tKSU = 12, tKHD = 6, tR = 2;
    specparam tPD = 20, tRST = 30, tW = 5;
    specparam tLH = 30, tHL = 27;
    $setuphold(posedge CLOCK &&& RESET, J, tJSU, tJHD, NOTIFIER);
    $setuphold(posedge CLOCK &&& RESET, K, tKSU, tKHD, NOTIFIER);
    $setuphold(posedge CLOCK &&& PRESET, J, tJSU, tJHD, NOTIFIER);
    $setuphold(posedge CLOCK &&& PRESET, K, tKSU, tKHD, NOTIFIER);
    $recovery(posedge RESET, posedge CLOCK, tR, NOTIFIER);
    $recovery(posedge PRESET, posedge CLOCK, tR, NOTIFIER);
    $period(posedge CLOCK, tPD, NOTIFIER);
    $width(posedge CLOCK, tW, tW/10, NOTIFIER);
    (J => Q) = tHL;
    (K => Q) = tLH;
    endspecify
endmodule
FIGURE 11.21 JK flipflop model with asynchronous preset and reset inputs


primitive JKFF_logic (Q, CLOCK, J, K, PRESET, RESET, NOTIFIER);
  output Q;
  reg Q; 
  input  CLOCK, J, K, PRESET, RESET, NOTIFIER;
 table	
  // CLK  J   K   PRE  RST  NO :  Q  :  Q+
      ?   ?   ?   0    1    ?  :  ?  :  1  ; //Async. preset 
      ?   ?   ?   *    1    ?  :  1  :  1  ; 
      ?   ?   ?   1    0    ?  :  ?  :  0  ; //Async. reset
      ?   ?   ?   1    *    ?  :  0  :  0  ; 
      r   0   0   1    1    ?  :  ?  :  -  ; //Hold on {JK} = 00
      r   0   1   1    1    ?  :  ?  :  0  ; //Reset on J = 0, K = 1
      r   1   0   1    1    ?  :  ?  :  1  ; //Set on J = 1, K = 0
      r   1   1   1    1    ?  :  0  :  1  ; //Toggle on {JK} = 11 
      r   1   1   1    1    ?  :  1  :  0  ; 
      f   ?   ?   ?    ?    ?  :  ?  :  -  ; //Ignore negative edge of clock
      b   *   ?   ?    ?    ?  :  ?  :  -  ; //Ignore data changes on steady clock
      b   ?   *   ?    ?    ?  :  ?  :  -  ;
      ?   ?   ?   ?    ?    *  :  ?  :  x  ; //Go unknown on any timing violation
endtable
endprimitive	
FIGURE 11.22 UDP model for JK flipflop


module DFF_behav(CLOCK, D, Q);
  input CLOCK, D;
  output Q;
  reg Q;
  always @(posedge CLOCK) Q <= D;
[bookmark: _GoBack]endmodule
FIGURE 11.23 Behavioral model of a simple D flipflop
