
/*ASIC style: asynchronous active low reset and preset inputs
EN is a synchronous logic signal*/

module ASIC_P_R(input CLK, RST, PRE, EN, D, output logic Q);
  always_ff @(posedge CLK, negedge RST, negedge PRE)
    if (!RST) Q <= 1'b0;
    else if (!PRE) Q <= 1'b1;
    else if (EN) Q <= D;
    else Q <= Q;
endmodule
FIGURE 9.1 ASIC coding style for a D flipflop with asynchronous preset and reset inputs


/*FPGA style: for optimal resource usage, always code
registers so that reset overrides preset and both have 
priority over clock enable (CE).

Preset and reset are active HIGH in this example. ASIC
style almost always has asynchronous preset and reset
active low.

No final "else" clause needed. If the clock is not
enabled, no action will occur.*/

module FPGA_P_R(input CLK, RST, PRE, CE, D, output logic Q);
  always_ff @(posedge CLK)
    if (RST) Q <= 1'b0;
    else if (PRE) Q <= 1'b1;
    else if (CE) Q <= D;
endmodule
FIGURE 9.2 FPGA style flipflop




module IRQ_ADDR1(input [7:0] IRQ, input [15:0] BASE_ADDR, 
  output logic [15:0] ADDR);
  parameter logic [15:0] OFFSET [0 : 7] = {4, 8, 12, 16, 20, 24, 28, 32};
  always_comb begin
    ADDR = BASE_ADDR;
    for (int I = 0; I < 8; I++)
      //Eight adders will be made
      if (IRQ[I]) ADDR = BASE_ADDR + OFFSET[I];
  end
endmodule

module IRQ_ADDR2(input [7:0] IRQ, input [15:0] BASE_ADDR, 
  output logic [15:0] ADDR);
  parameter logic [15:0] OFFSET [0 : 7] = {4, 8, 12, 16, 20, 24, 28, 32};
  logic [15:0] TEMP;
  always_comb begin
    TEMP = '0;
    for (int I = 0; I < 8; I++)
      if (IRQ[I]) TEMP = OFFSET[I];
      //This is NOT part of the "for" loop: Only one adder will be made
      ADDR = BASE_ADDR + TEMP;
  end
endmodule
FIGURE 9.13 Minimizing logic in a loop prevents redundant circuitry from being created





LOOP		LDR	COEF, [COEF_ADDR], #4 	;Fetch coefficient
		LDR	DATA, [DATA_ADDR], #4 	;Fetch datum
		SMLAL	PROD0, PROD1, COEF, DATA	;Signed long multiply-accumulate
		STR	PROD0, [RESULT_ADDR], #4	;Store lower word
		STR	PROD1, [RESULT_ADDR], #4	;Store upper word
		SUBS	COUNT, COUNT, #1		;Decrement loop
		BNE	LOOP				;Repeat if not done
FIGURE 9.20 ARM assembly language for a multiply accumulate loop




module share_adder #(SIZE = 4) (input CLOCK, SEL, 
  input [SIZE - 1 : 0] A, B, C, D,
  output logic [SIZE : 0] SUM);
  
  always_ff @(posedge CLOCK)
    if (SEL) SUM <= A + B;
    else SUM <= C + D;
      
endmodule

module dont_share_adder #(SIZE = 4) (input CLOCK, SEL, 
  input [SIZE - 1 : 0] A, B, C, D,
  output logic [SIZE : 0] SUM);
  logic [SIZE : 0] SUM0, SUM1;
  
  always_ff @(posedge CLOCK)
   if (SEL) SUM <= SUM0;
   else SUM <= SUM1;
     
  always_comb SUM0 = A + B;
  always_comb SUM1 = C + D;
endmodule
FIGURE 9.21 Resource sharing makes smaller but slower circuitry



module count_with_enable #(WIDTH = 4) (input CLK, RST, EN,
  output logic [WIDTH - 1 : 0] COUNT);
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) COUNT <= 0;
    else
      if (EN) COUNT <= COUNT + 1;
      else COUNT <= COUNT;
endmodule
FIGURE 9.25 Scalable counter with enable, a good candidate for clock gating when WIDTH > 2



